
	#include "WiFiS3.h"
#include "Arduino_LED_Matrix.h"
#include "Secrets.h" // <--- Archivo con la SSID y la clave para la red WiFi
#include "HTMLcode.h" // <--- Archivo con el HTML para la web

char ssid[] = SECRET_SSID;
char pass[] = SECRET_PASS;
int status = WL_IDLE_STATUS; //Estado de la WiFi, inicialmente en reposo
WiFiServer server(80); //Objeto de tipo sevidor (web)
ArduinoLEDMatrix matrix; //Objeto de tipo matriz de LED 8x12 (filas x columnas) integrada en la Arduino

byte frame_perlas[8][12] = { //Matriz de valores binarios de cada LED para las Perlas de Baily
  { 0, 0, 0, 0, 1, 1, 1, 1, 0, 0, 0, 0 },
  { 0, 0, 1, 1, 0, 0, 0, 0, 1, 1, 0, 0 },
  { 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0 },
  { 1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1 },
  { 1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1 },
  { 0, 1, 0, 0, 0, 0, 0, 0, 0, 0, 1, 0 },
  { 0, 0, 1, 1, 0, 0, 0, 0, 1, 1, 0, 0 },
  { 0, 0, 0, 0, 1, 1, 1, 1, 0, 0, 0, 0 }
};

//Calibración de los valores mínimo y máximo que da el sensor de luz
int luzMin = 13;
int luzMax = 911;

void setup() {
  Serial.begin(9600); //Arranca el Monitor Serie
  matrix.begin(); // Arranca la matriz de LED

  if (WiFi.status() == WL_NO_MODULE) { //Verifica disponibilidad del módulo WiFi de la Arduino
    Serial.println("Error WiFi.");
    while (true);
  }

  while (status != WL_CONNECTED) { //Conecta la Arduino a la WiFi
    Serial.print("Conectando a: ");
    Serial.println(ssid);
    status = WiFi.begin(ssid, pass);
    delay(5000); //Espera a que se complete el protocolo de conexión a la WiFi
  }
  server.begin(); //Arranca el servidor web de la Arduino
  printWifiStatus(); //Función que muestra el estado y la IP de la conexión, ver código más abajo
}

void loop() {
  WiFiClient client = server.available(); //El servidor web (server) atiende las peticiones de servicio (client)
  if (client) {
    // Leemos solo la primera línea de la petición (la más importante)
    String request = client.readStringUntil('\r');
    client.flush(); // Limpiamos el buffer para evitar saturación Rx

    if (request.indexOf("GET /data") != -1) { // Comprobación de si hay petición de servicio
      int val = analogRead(A0); //Lectura del
      int porcen = map(val, luzMin, luzMax, 0, 100); // Mapeo del valor de la escala del sensor a escala de 0% a 100%
      porcen = constrain(porcen, 0, 100); // Acotación del porcentaje para que no rebase los límites 0% y 100%

      client.println("HTTP/1.1 200 OK");
      client.println("Content-Type: text/plain");
      client.println("Connection: close");
      client.println();
      client.print(porcen); // Enviamos al cliente el valor del porcentaje obtenido

      if(porcen < 10) matrix.renderBitmap(frame_perlas, 8, 12);
      else matrix.clear(); // Para porcentaje <10 mostrar anillo de Perlas Baily en la Arduino, otro caso apagar la matriz LED

    } else {
      enviarPagina(client); // Envía la página al cliente, ver código más abajo
    }
    delay(10); // Respiro para el chip WiFi
    client.stop(); // Cierra la comunicación con el cliente
  }
}

void enviarPagina(WiFiClient& client) {
  client.println("HTTP/1.1 200 OK");
  client.println("Content-Type: text/html");
  client.println("Connection: close"); // Vital para evitar que se cuelgue la placa Arduino
  client.println();
  client.print(WEB_PAGE); // <--- Carga el contenido del archivo HTMLcode.h
}

void printWifiStatus() {
  IPAddress ip = WiFi.localIP();
  Serial.println("\n--- SERVIDOR LISTO ---");
  Serial.print("Escribe esto en tu navegador: http://");
  Serial.println(ip); // Proporciona la dirección IP para ver el gemelo digital en el navegador
  Serial.println("----------------------\n");
}



